

ソフトウェアサイエンス実験 S8 最終レポート

200911434 青木大祐

2012 年 11 月 15 日

1 概要

本実験では、関数プログラミング言語 Ocaml を用いて、Ocaml 自身のサブセットである「ミニ Ocaml 言語 (以下「ミニ Ocaml」)」のインタプリタおよびコンパイラを作成した。

2 Ocaml について

まず最初に、実験に用いる Ocaml について理解を深めるため、関数プログラミングの技法を学んだ。具体的には、再帰やパターンマッチを用いて Euclid の互除法から最大公約数を求める関数や、Fibonacci 数を求める関数を記述した。例を以下に示す。

```
1 let rec gcd (x, y) =
2   if x <= 0 || y <= 0 then gcd (abs x, abs y)
3   else if x = y then x
4   else if x > y then gcd(x-y, y)
5   else gcd(x, y-x);;
6
7 let rec fib n =
8   match n with
9     | 1 -> 1
10    | 2 -> 1
11    | n -> fib(n - 2) + fib(n - 1);;
```

3 ミニ Ocaml 言語

前章の内容を終えた後、本題である「ミニ Ocaml インタプリタの実装」に着手した。本実験で扱うミニ Ocaml の構文は以下のとおり^{*1}である。

1	<code>e ::= x</code>	変数
2	<code> let x = e in e</code>	let式
3	<code> let rec f x = e in e</code>	let-rec式
4	<code> fun x -> e</code>	関数
5	<code> e e</code>	関数適用
6	<code> true</code>	真理値リテラル(定数)
7	<code> false</code>	真理値リテラル(定数)
8	<code> if e then e else e</code>	if式
9	<code> n</code>	自然数リテラル (n は自然数 0, 1, 2, ...)
10	<code> - e</code>	整数演算(符号の反転)
11	<code> e + e</code>	整数演算(足し算)
12	<code> e * e</code>	整数演算(かけ算)
13	<code> e / e</code>	整数演算(割り算)
14	<code> e = e</code>	等しさ(整数と真理値)
15	<code> e > e</code>	大小比較(整数)
16	<code> e < e</code>	大小比較(整数)

このミニ Ocaml を実装するにあたって、入力されるプログラムを扱う内部的な表現として式 (*exp*) という構造を考える。プログラムは式からなり、また式は一つないしは複数の部分的な式から成り立っている。この再帰的な構造を表現するため、Ocaml のヴァリアント型を用いて代数的データ構造を実現することにする。

上記のミニ Ocaml の仕様を満たすために、次のような式の種類を *Syntax.ml* として用いる。なお、これは講義ページで配布されたものをそのまま利用するため、上記の仕様を含む、より多種の式の定義となっている。

```
1 type exp =
2   | Var of string          (* variable e.g. x *)
3   | IntLit of int          (* integer literal e.g. 17 *)
4   | BoolLit of bool       (* boolean literal e.g. true *)
5   | If of exp * exp * exp  (* if e then e else e *)
6   | Let of string * exp * exp (* let x=e in e *)
7   | LetRec of string * string * exp * exp (* letrec f x=e in e *)
8   | Fun of string * exp    (* fun x -> e *)
9   | App of exp * exp       (* function application i.e. e e *)
10  | Eq of exp * exp        (* e = e *)
11  | Noteq of exp * exp     (* e <> e *)
12  | Greater of exp * exp   (* e > e *)
13  | Less of exp * exp      (* e < e *)
14  | Plus of exp * exp      (* e + e *)
15  | Minus of exp * exp     (* e - e *)
16  | Times of exp * exp     (* e * e *)
```

^{*1} 実験テキスト (<http://logic.cs.tsukuba.ac.jp/kam/jikken/mini.html>) より引用

```

17 | Div of exp * exp      (* e / e *)
18 | Empty                (* [ ] *)
19 | Match of exp * ((exp * exp) list) (* match e with e->e | ... *)
20 | Cons of exp * exp     (* e :: e *)
21 | Head of exp           (* List.hd e *)
22 | Tail of exp           (* List.tl e *)

```

これらを踏まえて、与えられた式を評価して、最終的な値を計算するインタプリタを実装する。

3.1 eval 関数

式を受け取り、式の種類によって違う処理を行うため、Ocaml のパターンマッチングを用いて *eval* 関数を実装する。整数値の足し算と掛け算を評価する *eval1* を示す。

```

1 let rec eval1 e =
2   match e with
3   | IntLit(n)    -> n
4   | Plus(e1,e2)  -> (eval1 e1) + (eval1 e2)
5   | Times(e1,e2) -> (eval1 e1) * (eval1 e2)
6   | _           -> failwith "unknown expression"

```

もし与えられた式が整数リテラル (*IntLit*) だった場合は、その数値が式を評価した値となる。また、与えられた式が足し算 (*Plus*) だった場合は、2 つの引数として与えられた式を再帰的に評価し、その 2 値を合計し、評価した値として返す。掛け算についても同様である。

もし *exp* 型の値であるが *IntLit*, *Plus*, *Times* でないものが渡された場合、最後のパターン (.) にマッチし、*unknown expression* という例外が投げられる。

この *eval1* に、入力として $2*2+(3+(-4))$ という式を与えてみる。*exp* 型の表現では *Plus(Times(IntLit 2, IntLit 2), Plus(IntLit 3, IntLit (-4)))* である。実行結果を以下に示す。

```

1 val easy : exp =
2   Plus (Times (IntLit 2, IntLit 2), Plus (IntLit 3, IntLit (-4)))
3 - : int = 3

```

正しく計算できていることが分かる。

3.2 値と型

eval1 では評価した値は Ocaml の *integer* 型であるが、ミニ Ocaml のデータ型としてはまず *Int* 型と *Bool* 型を扱いたいため、このままでは不十分である。複数のデータ型を扱えるようにするため、値 (*value*) というデータ構造を導入する。

```

1 type value =
2   | IntVal of int      (* integer value e.g. 17 *)
3   | BoolVal of bool    (* boolean value e.g. true *)

```

式を評価すると値が返る。整数型が求められる式上の位置に真偽値型がある場合は型エラーであり、またその逆も同じくエラーである。例えば、条件節の真偽によって実行する処理を分岐する *If* 文では、条件節は真偽値型でなければならない。

以下に、*eval1* の式に加えて真偽値によって分岐する *If*、式 1 と式 2 が同じ値なら真を返す *Eq*、式 1 が式 2 よりも大きい整数であれば真を返す *Greater* を実装した *eval2* を以下に示す。

```

1 let rec eval2 e =
2   match e with
3   | IntLit(n) -> IntVal(n)
4   | Plus(e1,e2) ->
5     begin
6       match (eval2 e1, eval2 e2) with
7       | (IntVal(n1),IntVal(n2)) -> IntVal(n1+n2)
8       | _ -> failwith "integer values expected"
9     end
10  | Times(e1,e2) ->
11    begin
12      match (eval2 e1, eval2 e2) with
13      | (IntVal(n1),IntVal(n2)) -> IntVal(n1*n2)
14      | _ -> failwith "integer values expected"
15    end
16  | Eq(e1,e2) ->
17    begin
18      match (eval2 e1, eval2 e2) with
19      | (IntVal(n1),IntVal(n2)) -> BoolVal(n1=n2)

```

```

20 | (BoolVal(b1),BoolVal(b2)) -> BoolVal(b1=b2)
21 | _ -> failwith "wrong value"
22 end
23 | BoolLit(b) -> BoolVal(b)
24 | If(e1,e2,e3) ->
25   begin
26   match (eval2 e1) with
27   | BoolVal(true) -> eval2 e2
28   | BoolVal(false) -> eval2 e3
29   | _ -> failwith "wrong value"
30   end
31 | Greater (e1, e2) ->
32   begin
33   match (eval2 e1, eval2 e2) with
34   | (IntVal(n1), IntVal(n2)) -> BoolVal(n1 > n2)
35   | _ -> failwith "wrong value"
36   end
37 | _ -> failwith "unknown expression e";;

```

このままでは *Plus* や *Times* の定義の大部分が重複しており、また後から二項の整数演算を追加する際に煩雑になってしまう。これを回避するため、実際の計算を *binop* 関数に切り分けることにする。*binop* 関数の定義は以下のとおり。

```

1 let binop f e1 e2 =
2   match (eval2b e1, eval2b e2) with
3   | (IntVal(n1),IntVal(n2)) -> IntVal(f n1 n2)
4   | _ -> failwith "integer values expected"

```

これに伴って、*Plus*、*Times* の定義を以下のように変更する。

```

1 ...
2   | Plus(e1,e2) -> binop (+) e1 e2
3   | Times(e1,e2) -> binop ( * ) e1 e2
4 ...

```

3.3 変数と環境

ミニ Ocaml において変数を実現するため、*let* 文の導入を試みる。変数を束ねておく環境として、名前と値のタプルをリストにした (*string * value*) *list* を考える。変数を宣言した際にはこの環境に名前と値を追加し、式の中に変数が出現した際にはこの環境から名前を探しだして値を取り出すことが出来る。

3.3.1 変数の定義・参照

環境に変数を追加する操作として、環境の拡張 *ext* という関数を実装する。この関数は、引数に変数名と値、そして現在の環境を受け取り、先頭に追加したものを返す。

```

1 let ext env x v = (x,v) :: env

```

また、現在の環境から変数を参照する関数として *lookup* を実装する。これは変数名と現在の環境を受け取り、対応する値を返す。

```

1 let rec lookup x env =
2   match env with
3   | [] -> failwith ("unbound variable: " ^ x)
4   | (y,v)::tl -> if x=y then v
5   | _ -> lookup x tl

```

想定どおりの環境が作られているか、以下のようなテストプログラムを実行して確認してみる。なお、*emptyenv* は空の環境を返す関数である。

```

1 let env = emptyenv();;
2 let env = ext env "x" (IntVal 1);;
3 let env = ext env "y" (BoolVal true);;
4 let env = ext env "z" (IntVal 5);;
5
6 (* 実行結果 *)
7 val env : 'a list = []
8 val env : (string * value) list = [("x", IntVal 1)]
9 val env : (string * value) list = [("y", BoolVal true); ("x", IntVal 1)]
10 val env : (string * value) list =
11   [("z", IntVal 5); ("y", BoolVal true); ("x", IntVal 1)]

```

正しく環境が作られていることが分かったので、変数名を参照してみる。

```

1 let y = lookup "y" env;; (* 存在する変数 *)

```

```

1 let w = lookup "w" env;; (*存在しない変数*)
2
3
4 (*実行結果*)
5 val y : value = BoolVal true
6 Exception: Failure "unbound variable: w".

```

存在する変数は正しく参照できており、また未定義の変数は例外になっている。

なお、この環境は定義される度に先頭に追加されていく。*lookup* は先頭から順に変数名の探索を行うため、同名の変数が存在する場合は最新の物が参照されるため、変数名の上書きが可能になる。

3.3.2 ミニ Ocaml への組み込み

実装した *ext*, *lookup* をミニ Ocaml インタプリタに組みこむことを考える。

まず、動作の理解を深めるため、環境の変化を順に表示するデバッグ関数 *string_of_env*, *print_env* を実装した。

```

1 let string_of_env env =
2   let string_of_val (e:value) : string =
3     match e with
4     | IntVal n -> string_of_int n
5     | BoolVal true -> "true"
6     | BoolVal false -> "false"
7   in
8   let rec internal (env:(string * value) list) =
9     match env with
10    | [] -> ""
11    | (name, v)::rest -> name ^ " = " ^ (string_of_val v) ^ ";" ^ (internal rest)
12  in
13  "[" ^ (internal env) ^ "];";
14
15 let print_env (env: (string * value) list) =
16   print_string( string_of_env env);;

```

これを踏まえて、変数を定義する式 *Let* と変数を参照する式 *Var* を追加した *eval3* を以下に示す。デバッグモードを有効にするには、オプション引数として *~mode : 1* を指定すると、環境の変化が逐一出力される。

```

1 let rec eval3 ?(mode=0) e env = (* env を引数に追加 *)
2   if mode != 0 then print_env env;
3
4   let binop f e1 e2 env = (* binop の中でも eval3 を呼ぶので env を追加 *)
5     match (eval3 e1 env ~mode, eval3 e2 env ~mode) with
6     | (IntVal(n1),IntVal(n2)) -> IntVal(f n1 n2)
7     | _ -> failwith "integer value expected"
8   in
9   match e with
10  | Var(x) -> lookup x env
11  | IntLit(n) -> IntVal(n)
12  | BoolLit(b) -> BoolVal(b)
13  | Plus(e1,e2) -> binop (+) e1 e2 env (* env を追加 *)
14  | Times(e1,e2) -> binop ( * ) e1 e2 env (* env を追加 *)
15  | Eq(e1,e2) ->
16    begin
17      match (eval3 e1 env ~mode, eval3 e2 env ~mode) with
18      | (IntVal(n1),IntVal(n2)) -> BoolVal(n1=n2)
19      | (BoolVal(b1),BoolVal(b2)) -> BoolVal(b1=b2)
20      | _ -> failwith "wrong value"
21    end
22  | If(e1,e2,e3) ->
23    begin
24      match (eval3 e1 env ~mode) with
25      | BoolVal(true) -> eval3 e2 env (* env を追加 *)
26      | BoolVal(false) -> eval3 e3 env (* env を追加 *)
27      | _ -> failwith "wrong value"
28    end
29  | Let(x,e1,e2) ->
30    let env1 = ext env x (eval3 e1 env ~mode)
31    in eval3 e2 env1 ~mode
32  | _ -> failwith "unknown expression";;

```

テストとして、幾つかの入力を与えてみる。

```

1 (* let x = 1 in 2 + x *)
2 eval3 (Let ("x", IntLit 1, (Plus (IntLit 2, Var "x")))) [];;
3
4 (* let x = 1 in let y = 3 in x + y *)
5 eval3 (Let ("x", IntLit 1, Let("y", IntLit 3, (Plus (Var "y", Var "x"))))) [];;
6
7 (* let x = true in if x = true then 1 else 2 *)
8 eval3 (Let ("x", BoolLit true, If(Eq(Var "x", BoolLit true), IntLit 1, IntLit 2))) [];;
9

```

```

10 (* 実行結果 *)
11 - : value = IntVal 3
12 - : value = IntVal 4
13 - : value = IntVal 1

```

また、デバッグモードを用いて環境の遷移を表示してみる。

```

1 (* let x = 1 in let y = x + 1 in x + y *)
2 eval3 ~mode:1 (Let ("x", IntLit 1, (Let ("y", Plus(Var "x", IntLit 1), Plus(Var "x", Var "y"))))) [];;
3
4 [[]][x = 1;][x = 1;][x = 1;][x = 1;][y = 2;x = 1;][y = 2;x = 1;][y = 2;x = 1;]
5 - : value = IntVal 3

```

まず最初に x が定義され、次に y が定義されていく過程がわかる。

4 構文解析

今までは *exp* 型の式を自分で作成していたが、*ocamllex* と *ocamlyacc* を用いて、ミニ Ocaml 言語の構文を解析して *exp* 型の式を作り出すパーザを生成する。parser と lexer は実験にあたって与えられた *parser.mly* と *lexer.mll* を用いる。また、実行ファイルの生成にあたって必要となる他のファイル (*main.ml*, *syntax.ml*) も、与えられたものを用いることにする。

4.1 構文の拡張

lexer と parser を拡張して、新たに演算子を定義してみる。例として、二数が等しくないと真を返す演算子 $<>$ を定義する。

lexer.mll に新しいトークンとして $<>$ を *NOTEQUAL* という名前で追加する。

```

1 ...
2
3 (* 演算子 *)
4 | '+' { PLUS }
5 | '-' { MINUS }
6 | '*' { ASTERISK }
7 | '/' { SLASH }
8 | '=' { EQUAL }
9 | '<' { LESS }
10 | '>' { GREATER }
11 | ":::" { COLCOL }
12 | "<>" { NOTEQUAL }
13
14 ...

```

また、*parser.mly* を以下のように変更し、*NOTEQUAL* が出現した場合の処理として *Noteq* という *exp* を出力するように定義する。

```

1 ...
2
3 // 演算子
4 %token PLUS // '+'
5 %token MINUS // '-'
6 %token ASTERISK // '*'
7 %token SLASH // '/'
8 %token EQUAL // '='
9 %token LESS // '<'
10 %token GREATER // '>'
11 %token COLCOL // ":::"
12 %token NOTEQUAL // "<>"
13
14 ...
15
16 %left EQUAL GREATER LESS NOTEQUAL
17
18 ...
19
20 // e1 <> e2
21 | exp NOTEQUAL exp
22 { Noteq ($1, $3) }
23
24 ...

```

最後に、*syntax.ml* に *Noteq* という *exp* を追加する。

```

1 ...
2
3 (* 式の型 *)
4 type exp =

```

```

5 | Var of string          (* variable e.g. x *)
6 | IntLit of int          (* integer literal e.g. 17 *)
7 | BoolLit of bool        (* boolean literal e.g. true *)
8 | If of exp * exp * exp  (* if e then e else e *)
9 | Let of string * exp * exp (* let x=e in e *)
10 | LetRec of string * string * exp * exp (* letrec f x=e in e *)
11 | Fun of string * exp    (* fun x -> e *)
12 | App of exp * exp       (* function application i.e. e e *)
13 | Eq of exp * exp        (* e = e *)
14 | Noteq of exp * exp     (* e <> e *)
15 | Greater of exp * exp   (* e > e *)
16 | Less of exp * exp      (* e < e *)
17 | Plus of exp * exp      (* e + e *)
18 | Minus of exp * exp     (* e - e *)
19 | Times of exp * exp     (* e * e *)
20 | Div of exp * exp       (* e / e *)
21 | Empty                  (* [ ] *)
22 | Match of exp * ((exp * exp) list) (* match e with e->e | ... *)
23 | Cons of exp * exp      (* e :: e *)
24 | Head of exp            (* List.hd e *)
25 | Tail of exp            (* List.tl e *)
26
27 ...

```

これらを踏まえて、*eval* に *Noteq* の処理を記述する。

```

1 | Noteq(e1, e2) ->
2   begin
3   match (eval3 e1 env `mode, eval3 e2 env `mode) with
4   | (IntVal(n1),IntVal(n2)) -> BoolVal(n1!=n2)
5   | (BoolVal(b1),BoolVal(b2)) -> BoolVal(b1!=b2)
6   | _ -> failwith "wrong value"
7   end

```

以下のような入力を与えて動作を確認してみる。

```

1 run "true <> false";;
2 run "(1+1) <> 2";;
3 run "let x = 1 in let y = 2 in x <> y";;
4
5 - : Syntax.value = Syntax.BoolVal true
6 - : Syntax.value = Syntax.BoolVal false
7 - : Syntax.value = Syntax.BoolVal true

```

正しく動いていることが分かる。

4.2 関数と束縛

ミニ Ocaml において関数を実現するために、関数型という値を導入する。関数は静的束縛を実現するため、名前と関数本体に加えて関数抽象が行われた時点での環境を同時に保持する。

これに伴って、*value* 型の定義を以下のように拡張する。

```

1 type value =
2   | IntVal of int
3   | BoolVal of bool
4   | FunVal of string * exp * ((string * value) list)

```

また、関数抽象を行う式 *Fun* と関数適用を行う式 *App* の処理を以下のように定義する。

```

1 ...
2
3 | Fun(x,e1) -> FunVal(x, e1, env)
4 | App(e1,e2) ->
5   begin
6   match (eval4 e1 env) with
7   | FunVal(x,body,env1) ->
8     let arg = (eval4 e2 env)
9     in eval4 body (ext env1 x arg)
10  | _ -> failwith "function value expected"
11  end
12
13 ...

```

4.3 再帰関数

関数の再帰を実現するため、「再帰しない関数」とは別に再帰関数という型を用意する。型の定義は以下のとおり。

```
| RecFunVal of string * string * exp * ((string * value) list)
```

再帰関数を定義する式 *LetRec* と、関数適用 *App* に再帰関数を受け取った場合の処理を追加し、*eval6* を作成する。これで、ミニ Ocaml インタプリタの基本的な機能の実装は終了とする。

以下に、完成した *eval6* を含むミニ Ocaml インタプリタの全体を示す。

```
1 open Syntax;;
2
3 let emptyenv () = [];;
4
5 let ext env x v = (x,v) :: env;;
6
7 let rec lookup x env =
8   match env with
9   | [] -> failwith ("unbound variable: " ^ x)
10  | (y,v)::tl -> if x=y then v
11                  else lookup x tl;;
12
13 let string_of_env env =
14   let string_of_val e =
15     match e with
16     | IntVal n -> string_of_int n
17     | BoolVal true -> "true"
18     | BoolVal false -> "false"
19   in
20   let rec internal env =
21     match env with
22     | [] -> ""
23     | (name, v)::rest -> name ^ " = " ^ (string_of_val v) ^ ";" ^ (internal rest)
24   in
25   "[" ^ (internal env) ^ ""];;
26
27 (* eval3 : exp -> (string * value) list -> value *)
28 (* let と変数、環境の導入 *)
29 let print_env env =
30   print_string(string_of_env env);;
31
32 let rec eval6 e env = (* env を引数に追加 *)
33   let binop f e1 e2 env = (* binop の中でも eval6 を呼ぶので env を追加 *)
34     match (eval6 e1 env , eval6 e2 env ) with
35     | (IntVal(n1),IntVal(n2)) -> IntVal(f n1 n2)
36     | _ -> failwith "integer value expected"
37   in
38   match e with
39   | Var(x) -> lookup x env
40   | IntLit(n) -> IntVal(n)
41   | BoolLit(b) -> BoolVal(b)
42   | Plus(e1,e2) -> binop (+) e1 e2 env (* env を追加 *)
43   | Times(e1,e2) -> binop ( * ) e1 e2 env (* env を追加 *)
44   | Minus(e1, e2) -> binop (-) e1 e2 env
45   | Eq(e1,e2) ->
46     begin
47       match (eval6 e1 env , eval6 e2 env ) with
48       | (IntVal(n1),IntVal(n2)) -> BoolVal(n1=n2)
49       | (BoolVal(b1),BoolVal(b2)) -> BoolVal(b1=b2)
50       | _ -> failwith "wrong value"
51     end
52   | If(e1,e2,e3) ->
53     begin
54       match (eval6 e1 env ) with
55       | BoolVal(true) -> eval6 e2 env (* env を追加 *)
56       | BoolVal(false) -> eval6 e3 env (* env を追加 *)
57       | _ -> failwith "wrong value"
58     end
59   | Let(x,e1,e2) ->
60     let env1 = ext env x (eval6 e1 env )
61     in eval6 e2 env1
62   | LetRec(f,x,e1,e2) ->
63     let env1 = ext env f (RecFunVal (f, x, e1, env))
64     in eval6 e2 env1
65   | Noteq(e1, e2) ->
66     begin
67       match (eval6 e1 env , eval6 e2 env ) with
68       | (IntVal(n1),IntVal(n2)) -> BoolVal(n1!=n2)
69       | (BoolVal(b1),BoolVal(b2)) -> BoolVal(b1!=b2)
70       | _ -> failwith "wrong value"
71     end
72   | Fun(x,e1) -> FunVal(x, e1, env)
73   | App(e1,e2) ->
74     let funpart = (eval6 e1 env) in
75     let arg = (eval6 e2 env) in
76     begin
77       match funpart with
78       | FunVal(x,body,env1) ->
```



```

79         let env2 = (ext env1 x arg) in
80         eval6 body env2
81       | RecFunVal(f,x,body,env1) ->
82         let env2 = (ext (ext env1 x arg) f funpart) in
83         eval6 body env2
84       | _ -> failwith "wrong value in App"
85     end
86 | Greater(e1, e2) ->
87   begin
88     match (eval6 e1 env , eval6 e2 env ) with
89     | (IntVal(n1), IntVal(n2)) -> BoolVal(n1 > n2)
90     | _ -> failwith "wrong value"
91   end
92 | _ -> failwith "unknown expression";;

```

動作確認として、自然数 x について階乗を求める再帰関数 *fact* をミニ Ocaml で実装し、実行した。結果は以下のとおり。

```

1 let run src =
2   Eval.eval6 (Main.parse(src)) (Eval.emptyenv())
3 ;;
4
5 run "let rec fact x = if x = 0 then 1 else x * fact (x-1) in fact 10";;

```

```

1 # val run : string -> Syntax.value = <fun>
2 - : Syntax.value = Syntax.IntVal 3628800

```

5 ミニ Ocaml コンパイラの作成

ここまで作成したミニ Ocaml インタプリタをもとに、仮想機械 ASEC の命令列を出力するコンパイラを作成する。

5.1 ASEC 機械

ASEC 機械とは、*Accumulator, Stack, Environment, Code* の 4 部分で構成される仮想機械である。資料として提示された ASEC 仮想機械の遷移表と命令列の変換規則にしたがって、今まで製作したインタプリタをもとにコンパイラを作成した。

以下に、コンパイラとして製作した *eval* 関数を含む *eval.ml* と、課題として ASEC 機械に機能追加を行った *am.ml* を示す。

ソースコード 1 eval.ml

```

1 open Syntax;;
2 open Am;;
3
4 let emptyenv () = [];;
5
6 let rec position (x : string) (venv : string list) : int =
7   match venv with
8   | [] -> failwith "no matching variable in environment"
9   | y::venv2 -> if x=y then 0 else (position x venv2) + 1;;
10
11 let rec eval e env =
12   (* env を引数に追加 *)
13   let binop f e1 e2 env =
14     (* binop の中でも eval を呼ぶので env を追加 *)
15     (eval e2 env) @ [I_Push] @ (eval e1 env) @ [f]
16   in
17   match e with
18   | Var(x) -> [I_Search (position x env)]
19   | IntLit(n) -> [I_Ldi n]
20   | BoolLit(b) -> [I_Ldb b]
21   | Plus(e1,e2) -> binop I_Add e1 e2 env
22   | Minus(e1,e2) -> binop I_Sub e1 e2 env
23   | Times(e1,e2) -> binop I_Mult e1 e2 env
24   | Div(e1,e2) -> binop I_Div e1 e2 env
25   | Eq(e1,e2) -> binop I_Eq e1 e2 env
26   | Noteq(e1,e2) -> binop I_Noteq e1 e2 env
27   | If(e1,e2,e3) -> (eval e1 env) @ [I_Test ((eval e2 env), (eval e3 env))]
28   | Let(x,e1,e2) ->
29     let env1 = x::env in
30     [I_Pushenv] @ (eval e1 env) @ [I_Extend] @ (eval e2 env1) @ [I_Popenv]
31   | LetRec(f,x,e1,e2) ->
32     let env1 = f::env in
33     [I_Pushenv] @ [I_Mkclos (eval e1 (x::env1))] @ [I_Extend] @ (eval e2 env1) @ [I_Popenv]
34   | Fun(x,e1) -> [I_Mkclos (eval e1 (x::("_":env)))]
35   | App(e1,e2) -> [I_Pushenv] @ (eval e2 env) @ [I_Push] @ (eval e1 env) @ [I_Apply; I_Popenv]
36   | Greater(e1, e2) -> binop I_Greater e1 e2 env
37   | _ -> failwith "unknown expression";;

```

ソースコード 2 am.ml(抜粋)

```

1  ...
2  instr =
3  | I_Ldi of int          (* I_Ldi(n) は、整数nを Accumulator に置く (loadする) *)
4  | I_Ldb of bool        (* I_Ldb(b) は、真理値bを Accumulator に置く (loadする) *)
5  | I_Push               (* Accumulatorにある値をスタックに積む *)
6  | I_Extend             (* Accumulatorにある値を環境に積む (環境を拡張する) *)
7  | I_Search of int      (* I_Search(i) は、環境の i 番目の値を取ってきて Accumulatorに置く *)
8  | I_Pushenv            (* 現在の環境を、スタックに積む *)
9  | I_Popenv             (* スタックのトップにある環境を、現在の環境とする *)
10 | I_Mkclos of code      (* I_Mkclos(c) は、関数本体のコードが c で
11     * ある関数クロージャを生成し、Accumulatorに置く。
12     * なお、関数の引数は、変数名を除去しているので、
13     * このクロージャに含まれない。関数本体で
14     * は「環境の中の0番目の変数として指定される。
15     * ここでは、すべての関数を再帰関数として処理している。
16     *)
17 | I_Apply               (* Accumulator の値が関数クロージャである
18     * とき、その関数を、スタックトップにある
19     * 値に適用した計算を行なう。
20     *)
21 | I_Test of code * code (* I_Test(c1,c2)は、Accumulatorにある値が
22     * true ならば、コードc1 を実行し、false
23     * ならばコード c2 を実行する。
24     *)
25 | I_Add                 (* Accumulatorにある値にスタックトップの値
26     * を加えて結果をAccumulator にいれる
27     *)
28 | I_Sub                 (* Accumulatorにある値にスタックトップの値
29     * を引いて結果をAccumulatorにいれる *)
30 | I_Mult                (* Accumulatorにある値にスタックトップの値
31     * を掛けて結果をAccumulatorにいれる *)
32 | I_Div                 (* Accumulatorにある値にスタックトップの値
33     * で割って結果をAccumulatorにいれる *)
34 | I_Greater             (* Accumulatorにある値がスタックトップの値
35     * より大きいかどうかを比較して、結果をAccumulatorにいれる *)
36 | I_Eq                  (* Accumulatorにある値とスタックトップの値
37     * が同じ整数であるかどうかをテストして、結
38     * 果をAccumulator にいれる
39     *)
40 | I_Noteq                (* Accumulatorにある値とスタックトップの値
41     * が違う整数であるかどうかをテストして、結
42     * 果をAccumulator にいれる
43     *)
44 ...
45
46 (* ASEC machine の状態遷移(計算)*)
47 let rec trans (a:am_value) (s:am_stack) (e:am_env) (c:code) : am_value =
48   let binop (a:am_value) (s:am_stack) (f:instr) : (am_value * am_stack) =
49     begin
50       match (f, a, s) with
51       | (I_Add, AM_IntVal(n), AM_IntVal(m)::s1) -> (AM_IntVal (n+m), s1)
52       | (I_Sub, AM_IntVal(n), AM_IntVal(m)::s1) -> (AM_IntVal (n-m), s1)
53       | (I_Mult, AM_IntVal(n), AM_IntVal(m)::s1) -> (AM_IntVal (n*m), s1)
54       | (I_Div, AM_IntVal(n), AM_IntVal(m)::s1) -> (AM_IntVal (n/m), s1)
55       | (I_Greater, AM_IntVal(n), AM_IntVal(m)::s1) -> (AM_BoolVal (n>m), s1)
56       | (I_Eq, AM_IntVal(n), AM_IntVal(m)::s1) -> (AM_BoolVal (n=m), s1)
57       | (I_Noteq, AM_IntVal(n), AM_IntVal(m)::s1) -> (AM_BoolVal (n!=m), s1)
58       | _ -> failwith "unexpected type of argument(s) for binary operation"
59     end
60   in
61   match (a,s,e,c) with
62   | (_,[],[],[]) -> a (* コードが空の時、Accumulatorの値が最終結果 *)
63   | (_,_,_,[]) -> failwith "non-empty stack or environment"
64   (* 計算が終わった時、スタックと環境は空であるべき。 *)
65   | (_,_,_,I_Mkclos(c2)::c1) -> trans (AM_Closure(c2,e)) s e c1
66   | (_,_,_,I_Push ::c1) -> trans a (a::s) e c1
67   | (_,_,_,I_Extend ::c1) -> trans a s (a::e) c1
68   | (_,_,_,I_Search(n) ::c1) -> trans (List.nth e n) s e c1
69   | (_,_,_,I_Pushenv ::c1) -> trans a (AM_Env(e)::s) e c1
70   | (_,AM_Env(e1)::s1,_,I_Popenv::c1) -> trans a s1 e1 c1
71   | (_,_,_,I_Popenv::c1) -> failwith "non-environment on the stack top"
72   | (_,_,_,I_Popenv::c1) -> failwith "empty stack for Pop"
73   | (AM_Closure(c2,e1),v::s1,_,I_Apply::c1) -> trans a s1 (v::a::e1) (c2@c1)
74   | (_,_,_,I_Apply::c1) -> failwith "closure expected in accumulator for Apply"
75   | (_,_,_,I_Apply::c1) -> failwith "empty stack for Apply"
76   | (_,_,_,I_Ldi(n)::c1) -> trans (AM_IntVal(n)) s e c1
77   | (_,_,_,I_Ldb(b)::c1) -> trans (AM_BoolVal(b)) s e c1
78   | (AM_BoolVal(true),_,_,I_Test(c2,_)::c1) -> trans a s e (c2 @ c1)
79   | (AM_BoolVal(false),_,_,I_Test(_,c3)::c1) -> trans a s e (c3 @ c1)
80   | (_,_,_,I_Test(_,_)::c1) -> failwith "boolean expected for if-test"
81   | (_,_,_,i ::c1)
82   when List.mem i [I_Add; I_Sub; I_Mult; I_Div; I_Greater; I_Eq; I_Noteq]
83   ->
84     let (v,s1) = binop a s i
85     in trans v s1 e c1
86   | _ -> failwith "unknown instruction"
87
88 ...

```

```

1 let run src =
2   Eval.eval (Main.parse(src)) (Eval.emptyenv())
3 ;;
4
5
6 let instr = run "let rec fact x = if x = 0 then 1 else x * fact (x-1) in fact 10";;
7 Am.am_eval instr;;

```

```

1 # val run : string -> Am.code = <fun>
2 val instr : Am.code =
3   [Am.I_Pushenv;
4     Am.I_Mkclos
5     [Am.I_Ldi 0; Am.I_Push; Am.I_Search 0; Am.I_Eq;
6       Am.I_Test ([Am.I_Ldi 1],
7         [Am.I_Pushenv; Am.I_Ldi 1; Am.I_Push; Am.I_Search 0; Am.I_Sub;
8           Am.I_Push; Am.I_Search 1; Am.I_Apply; Am.I_Popenv; Am.I_Push;
9             Am.I_Search 0; Am.I_Mult]]]);
10    Am.I_Extend; Am.I_Pushenv; Am.I_Ldi 10; Am.I_Push; Am.I_Search 0;
11    Am.I_Apply; Am.I_Popenv; Am.I_Popenv]
12 - : Am.am_value = Am.AM_IntVal 3628800

```

正しく計算できていることが分かる。